
Cloak is an Elixir encryption library that implements several best practices
and conveniences for Elixir developers:
	Random IVs
	Tagged ciphertexts
	Elixir-native configuration

Documentation
	Hex Documentation (Includes installation guide)
	How to upgrade from Cloak 0.9.x to 1.0.x

Examples
Encrypt / Decrypt
{:ok, ciphertext} = MyApp.Vault.encrypt("plaintext")
# => {:ok, <<1, 10, 65, 69, 83, 46, 71, 67, 77, 46, 86, 49, 45, 1, 250, 221,
# =>  189, 64, 26, 214, 26, 147, 171, 101, 181, 158, 224, 117, 10, 254, 140, 207, 
# =>  215, 98, 208, 208, 174, 162, 33, 197, 179, 56, 236, 71, 81, 67, 85, 229, 
# =>  ...>>}

MyApp.Vault.decrypt(ciphertext)
# => {:ok, "plaintext"}
Reencrypt With New Algorithm/Key
"plaintext"
|> MyApp.Vault.encrypt!(:aes_256)
|> MyApp.Vault.decrypt!()
|> MyApp.Vault.encrypt!(:aes_256)
|> MyApp.Vault.decrypt!()
# => "plaintext"
Configuration
config :my_app, MyApp.Vault,
  ciphers: [
    # In AES.GCM, it is important to specify 12-byte IV length for
    # interoperability with other encryption software. See this GitHub issue
    # for more details: https://github.com/danielberkompas/cloak/issues/93
    # 
    # In Cloak 2.0, this will be the default iv length for AES.GCM.
    aes_gcm: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: <<...>>, iv_length: 12},
    aes_ctr: {Cloak.Ciphers.AES.CTR, tag: "AES.CTR.V1", key: <<...>>}
  ]
Features
Random Initialization Vectors (IV)
Every strong encryption algorithm recommends unique initialization vectors.
Cloak automatically generates unique vectors using
:crypto.strong_rand_bytes, and includes the IV in the ciphertext.
This greatly simplifies storage and is not a security risk.
Tagged Ciphertext
Each ciphertext contains metadata about the algorithm and key which was used
to encrypt it. This allows Cloak to automatically select the correct key and
algorithm to use for decryption for any given ciphertext.
This makes key rotation much easier, because you can easily tell whether any
given ciphertext is using the old key or the new key.
Elixir-Native Configuration
Cloak works through Vault modules which you define in your app, and add
to your supervision tree.
You can have as many vaults as you wish running simultaneously in your
project. (This works well with umbrella apps, or any runtime environment
where you have multiple OTP apps using Cloak)
Ecto Support
You can use Cloak to transparently encrypt Ecto fields, using
cloak_ecto.
Security Notes
	Cloak is built on Erlang's crypto library, and therefore inherits its security.
	You can implement your own cipher modules to use with Cloak, which may use any other encryption algorithms of your choice.

Setup
Define a vault module
This is the module you will use to encrypt/decrypt your ciphertext. You can have more than one!
defmodule MyApp.MyVault do
  use Cloak.Vault, otp_app: :my_app
end
Generate an encryption key
This will generate a 256-bit key in Base64 encoding that you can store in an environment variable.
32
|> :crypto.strong_rand_bytes()
|> Base.encode64
Custom Cipher Module
If you don't want to use any of the included cipher modules, you can define your own. See Cloak.Cipher.
defmodule MyApp.MyCipher do
  @behaviour Cloak.Cipher

  @impl true
  def encrypt(plaintext, opts) do
    # your logic here
  end

  @impl true
  def decrypt(ciphertext, opts) do
    # your logic here
  end

  @impl true
  def can_decrypt?(ciphertext, opts) do
    # return a boolean
  end
end
Configure
Using Config
This is the easiest option, and the vault will automatically read this configuration.
# config/runtime.exs
config :my_app, MyApp.MyVault,
    ciphers: [
      default: Cloak.Ciphers.AES.GCM, 
      tag: "AES.GCM.V1", 
      key: Base.decode64!("your-key-here"),
      iv_length: 12
    ]
Inside the Vault module
This is best if you need to fetch the key from some custom store before you use it. Add these lines to your vault module.
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app

  @impl GenServer
  def init(config) do
    config =
      Keyword.put(config, :ciphers, [
        default: {
          Cloak.Ciphers.AES.GCM, 
          tag: "AES.GCM.V1", 
          key: decode_env!("CLOAK_KEY"),
          iv_length: 12
        }
      ])

    {:ok, config}
  end

  defp decode_env!(var) do
    var
    |> System.get_env()
    |> Base.decode64!()
  end
end
Usage
Encrypt
Safe: return any errors
MyApp.MyVault.encrypt("plaintext")
# => {:ok, "ciphertext"}
Unsafe: raise any errors
MyApp.MyVault.encrypt!("plaintext")
# => "ciphertext"
Decrypt
Safe: return any errors
MyApp.MyVault.decrypt("ciphertext")
# => {:ok, "plaintext"}
Unsafe: raise any errors
MyApp.MyVault.decrypt!("ciphertext")
# => "plaintext"
Usage with Ecto
See the cloak_ecto library! Here's a preview:
defmodule MyApp.MySchema do
  use Ecto.Schema

  schema "table" do
    field :encrypted_field, MyApp.Encrypted.Binary
  end
end

This guide will walk you through installing Cloak in your project.
Add the Dependency
First, add :cloak to your dependencies in mix.exs:
{:cloak, "1.1.1"}
Run mix deps.get to fetch the dependency.
Generate a Key
You'll need a secret key for encryption. This is easy to generate in the
IEx console.
$ iex
iex> 32 |> :crypto.strong_rand_bytes() |> Base.encode64()
"aJ7HcM24BcyiwsAvRsa3EG3jcvaFWooyQJ+91OO7bRU="

This will generate a relatively strong encryption 256-bit encryption
key encoded with Base64.
Create a Vault
Next, create a Cloak.Vault for your project.
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app
end
Configure it as shown in the Cloak.Vault documentation, with at least one
active cipher. Note that the :key needs to be decoded from Base64 encoding into
its raw binary form.
config :my_app, MyApp.Vault,
  ciphers: [
    default: {
      Cloak.Ciphers.AES.GCM, 
      tag: "AES.GCM.V1", 
      key: Base.decode64!("your-key-here"),
      # In AES.GCM, it is important to specify 12-byte IV length for
      # interoperability with other encryption software. See this GitHub
      # issue for more details:
      # https://github.com/danielberkompas/cloak/issues/93
      # 
      # In Cloak 2.0, this will be the default iv length for AES.GCM.
      iv_length: 12
    }
  ]
If you want to fetch keys from system vars, you should use the init/1 callback
to configure the vault instead:
# Assumes that you have a CLOAK_KEY environment variable containing a key in
# Base64 encoding.
#
# export CLOAK_KEY="A7x+qcFD9yeRfl3GohiOFZM5bNCdHNu27B0Ozv8X4dE="

defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app

  @impl GenServer
  def init(config) do
    config =
      Keyword.put(config, :ciphers, [
        default: {
          Cloak.Ciphers.AES.GCM, 
          tag: "AES.GCM.V1", 
          key: decode_env!("CLOAK_KEY"),
          iv_length: 12
        }
      ])

    {:ok, config}
  end

  defp decode_env!(var) do
    var
    |> System.get_env()
    |> Base.decode64!()
  end
end
Finally, add your vault to your supervision tree.
children = [
  MyApp.Vault
]
Usage
You can now encrypt and decrypt values using your Vault.
{:ok, ciphertext} = MyApp.Vault.encrypt("plaintext") 
# => {:ok, <<1, 10, 65, 69, 83, 46, 71, 67, 77, 46, 86, 49, 93, 140, 255, 234,
1, 195, 125, 112, 121, 186, 169, 185, 129, 122, 237, 161, 160, 24, 166,
48, 224, 230, 53, 194, 251, 175, 215, 10, 186, 130, 61, 230, 176, 102,
213, 209, ...>>}

MyApp.Vault.decrypt(ciphertext)
{:ok, "plaintext"}
By default, the first configured key will be used. You can use a specific key
to use by referencing its label:
MyApp.Vault.encrypt("plaintext", :default)
Decryption will use the metadata embedded in the ciphertext to decide which
configured key to use.
Usage with Ecto
If you want to use Cloak to automatically encrypt and decrypt fields in your
Ecto schemas, see cloak_ecto.

The easiest way to generate a key is via IEx.
$ iex
iex> 32 |> :crypto.strong_rand_bytes() |> Base.encode64()
"HXCdm5z61eNgUpnXObJRv94k3JnKSrnfwppyb60nz6w="

This will generate a strong 256-bit key encoded with Base64.



  

    
How To Upgrade From 0.9.x to 1.0.x
    

Cloak 1.0.0 separated the Ecto components into a separate package:
cloak_ecto. This required some breaking name changes.
	Cloak's default JSON library is now Jason, not Poison
	All Ecto-related modules now have the prefix Cloak.Ecto.
	Cloak's Ecto types have a new naming scheme: Cloak.Fields.Binary became
Cloak.Ecto.Binary.
	mix cloak.migrate has been renamed to mix cloak.migrate.ecto.

Install cloak_ecto 1.0.0
Replace your cloak dependency with cloak_ecto.
{:cloak_ecto, "1.0.0"}
Configure JSON Library
If you have not specified the :json_library setting in your Vault, and want
to continue using Poison, you will need to add that setting:
config :my_app, MyApp.Vault,
  json_library: Poison,
  # ...
Otherwise, you will need to add Jason to your dependencies:
{:jason, "~> 1.0"}
Update Cloak Field References
Update all your Cloak field references from the old naming scheme to the new.
For example:
use Cloak.Fields.Binary, vault: MyApp.Vault
Should now become:
use Cloak.Ecto.Binary, vault: MyApp.Vault
Update Migration References
If you use Cloak.Migrator directly, rename your references to
Cloak.Ecto.Migrator instead. If you use mix cloak.migrate, rename any
references to mix cloak.migrate.ecto instead..



  

    
How To Upgrade From 0.8.x to 0.9.x
    

Cloak 0.9.0 encrypts and decrypts data up to 200% faster than 0.8.0.
This result is achieved by caching configuration data in an ETS table.
	Breaking Change: You must now add your Cloak.Vault to your
supervision tree.

Update Your Dependency
Update your cloak dependency to 0.9.0 or higher:
{:cloak, "~> 0.9.2"}
Supervise Your Vault
Add your vault to your supervision tree:
children = [
  MyApp.Vault
]
Update init/1 callback implementation
If you are fetching the keys from system vars, you should change the implementation of the init/1 callback to GenServer.init/1
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app

  @impl GenServer
  def init(config) do
    (...)
  end
end



  

    
How To Upgrade From 0.7.x to 0.8.x
    

Cloak 0.8 introduces a new feature with no breaking changes.
	mix cloak.migrate now supports binary primary keys. It iterates
through database tables using a cursor, instead of min/max ID.

	Cloak.CustomCursor allows you to customize fields used for the cursor.


Update Your Dependency
Update your cloak dependency to 0.8 or later.
{:cloak, "~> 0.8.0"}
Most users will not need to do anything more. See the Cloak.CustomCursor
docs for more details.

Cloak 0.7 introduced a number of important changes.
	Encryption is now performed through Cloak.Vault modules
	Ciphertext no longer contains a module_tag
	Ecto types are now local to your project
	You no longer need an :encryption_version field

Install Cloak 0.7
Update your cloak dependency to 0.7 or later.
{:cloak, "~> 0.7.0"}
Create a Vault
Create a vault module for your project. (Or more than one, if you want!)
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app
end
You will need to move your existing configuration to the vault. For example,
if you had this configuration:
config :cloak, Cloak.AES.CTR,
  tag: "AES",
  default: true,
  keys: [
    %{tag: <<1>>, key: :base64.decode("..."), default: true}
  ]
You would convert it to the following:
config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.CTR, tag: "AES.V2", key: Base.decode64!("...")},
    retired: {Cloak.Ciphers.Deprecated.AES.CTR, module_tag: "AES", tag: <<1>>, key: Base.decode64!("...")}
  ]
Notice that the tag: "AES" became module_tag: "AES" in the :retired
cipher configuration.
Alternatively, if your keys are stored in environment variables, you could
configure the vault using the init/1 callback:
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app

  @impl Cloak.Vault
  def init(config) do
    config =
      Keyword.put(config, :ciphers, [
        default: {Cloak.Ciphers.AES.CTR, tag: "AES.V2", key: decode_env("CLOAK_KEY")},
        retired: {Cloak.Ciphers.Deprecated.AES.CTR, module_tag: "AES", tag: <<1>>, key: decode_env("CLOAK_KEY")}
      ])

    {:ok, config}
  end

  defp decode_env(var) do
    var
    |> System.get_env()
    |> Base.decode64!()
  end
end
Create Project-Specific Ecto Types
For each type of encrypted field you have, define a local type. For example,
if you had the following schema:
defmodule MyApp.Accounts.User do
  use Ecto.Schema

  import Ecto.Changeset

  schema "users" do
    field :name, Cloak.EncryptedBinaryField,
    field :encryption_version
  end

  @doc false
  def changeset(struct, attrs \\ %{}) do
    struct
    |> cast(attrs, [:name])
    |> put_change(:encryption_version, Cloak.version())
  end
end
You would define a project-specific field:
defmodule MyApp.Encrypted.Binary do
  use Cloak.Fields.Binary, vault: MyApp.Vault
end
And then replace Cloak.EncryptedBinaryField in your schema:
schema "users" do
  field :name, MyApp.Encrypted.Binary,
  field :encryption_version
end
Finally, you'd remove the :encryption_version field as it is no longer
needed.
# In migration...
alter table(:users) do
  remove :encryption_version
end

# In your changeset...
@doc false
def changeset(struct, attrs \\ %{}) do
  struct
  |> cast(attrs, [:name])
end
Migrate Existing Data
To convert ciphertext en masse from the old v0.6 format to the new v0.7
format, you'll need to run mix cloak.migrate as shown in its documentation.
mix cloak.migrate -r MyApp.Repo -s MyApp.Schema
Remove :retired Cipher
Now that the data has been migrated to the new v0.7 format, you can remove the
:retired cipher from your configuration.
config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.CTR, tag: "AES.V2", key: Base.decode64!("...")}
  ]

Cloak consists of two behaviours:
	Cloak.Cipher - Ciphers encrypt and decrypt data using a particular
encryption algorithm and key.

	Cloak.Vault - Vaults configure and use ciphers. Applications can
have multiple vaults.

A behaviour for encryption/decryption modules. You can rely on this behaviour
to create your own Cloak-compatible cipher modules.
Example
We will create a cipher that simply prepends "Hello, " to any given
plaintext on encryption, and removes the prefix on decryption.
First, define your own cipher module, and specify the Cloak.Cipher
behaviour.
defmodule MyApp.PrefixCipher do
  @behaviour Cloak.Cipher
end
Add some configuration to your vault for this new cipher:
config :my_app, MyApp.Vault,
  ciphers: [
    prefix: {MyApp.PrefixCipher, prefix: "Hello, "}
  ]
The keyword list containing the :prefix will be passed in as opts
to our cipher callbacks. You should specify any options your cipher will
need for encryption/decryption here, such as the key.
Next, define the can_decrypt?/2 callback:
@impl true
def can_decrypt?(ciphertext, opts) do
  String.starts_with?(ciphertext, opts[:prefix])
end
If the ciphertext starts with "Hello, ", we know it was encrypted with this
cipher and we can proceed. Finally, define the encrypt and decrypt
functions:
@impl true
def encrypt(plaintext, opts) do
  opts[:prefix] <> plaintext
end

@impl true
def decrypt(ciphertext, opts) do
  String.replace(ciphertext, opts[:prefix], "")
end
You can now use your cipher with your vault!
MyApp.Vault.encrypt!("World!", :prefix)
# => "Hello, World!"

MyApp.Vault.decrypt!("Hello, World!")
# => "World!"

        can_decrypt?(ciphertext, opts)

      


        Determines if a given ciphertext can be decrypted by this cipher. Options
are derived from the cipher configuration. See encrypt/2.



    


    
      
        decrypt(ciphertext, opts)

      


        Decrypt a value, using the given opts. Options are derived from the cipher
configuration. See encrypt/2.

        Encrypt a value, using the given keyword list of options. These options

      Link to this callback
    
    can_decrypt?(ciphertext, opts)

          @callback can_decrypt?(ciphertext(), opts()) :: boolean()


      


Determines if a given ciphertext can be decrypted by this cipher. Options
are derived from the cipher configuration. See encrypt/2.

  



  
    
      
      Link to this callback

          @callback decrypt(ciphertext(), opts()) :: {:ok, binary()} | :error


      


Decrypt a value, using the given opts. Options are derived from the cipher
configuration. See encrypt/2.

  



  
    
      
      Link to this callback
    
    encrypt(plaintext, opts)

          @callback encrypt(plaintext(), opts()) :: {:ok, binary()} | :error


      


Encrypt a value, using the given keyword list of options. These options
derive from the cipher configuration, like so:
config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: <<1, 0, ...>>}
  ]
The above configuration will result in the following opts being passed
to this function:
[tag: "AES.GCM.V1", key: <<1, 0, ...>>]
Your implementation must include any information it will need for
decryption in the generated ciphertext.

  


        

      



  

    
Cloak.Vault behaviour
    



      
Encrypts and decrypts data, using a configured cipher.
Create Your Vault
Define a module in your application that uses Cloak.Vault.
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app
end
Configuration
The :otp_app option should point to an OTP application that has the vault
configuration.
For example, the vault:
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app
end
Could be configured with Mix configuration like so:
config :my_app, MyApp.Vault,
  json_library: Jason,
  ciphers: [
    default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: <<...>>}
  ]
The configuration options are:
	:json_library: Used to convert data types like lists and maps into
binary so that they can be encrypted. (Default: Jason)

	:ciphers: a list of Cloak.Cipher modules the following format:
  {:label, {CipherModule, opts}}
The first configured cipher in the list is the default for encrypting
all new data, regardless of its label. This behaviour can be overridden
on a field-by-field basis.
The opts are specific to each cipher module. Check their
codumentation for what each cipher requires.
	Cloak.Ciphers.AES.GCM
	Cloak.Ciphers.AES.CTR



Runtime Configuration
Because Vaults are GenServers, they can be configured at runtime using the
init/1 callback. This allows you to easily fetch values like environment
variables in a reliable way.
The configuration from the :otp_app is passed as the first argument to the
callback, allowing you to append to or change it at will.
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app

  @impl GenServer
  def init(config) do
    config =
      Keyword.put(config, :ciphers, [
        default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: decode_env!("CLOAK_KEY")}
      ])

    {:ok, config}
  end

  defp decode_env!(var) do
    var
    |> System.get_env()
    |> Base.decode64!()
  end
end
You can also pass configuration to vaults via start_link/1:
MyApp.Vault.start_link(ciphers: [
  default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: key}
])
Supervision
Because Vaults are GenServers, you'll need to add your vault to your
supervision tree in application.ex or whichever supervisor you prefer.
children = [
  MyApp.Vault
]
If you want to pass in configuration values at runtime, you can do so:
children = [
  {MyApp.Vault, ciphers: [...]}
]
Usage
You can use the vault directly by calling its functions.
MyApp.Vault.encrypt("plaintext")
# => {:ok, <<...>>}

MyApp.Vault.decrypt(ciphertext)
# => {:ok, "plaintext"}
See the documented callbacks below for the functions you can call.
Performance Notes
Vaults are not bottlenecks. They simply store configuration in an ETS table
named after the Vault, e.g. MyApp.Vault.Config. All encryption and
decryption is performed in your local process, reading configuration from
the vault's ETS table.

        decrypt(ciphertext)

      


        Decrypts a binary with the configured cipher that generated the binary.
Automatically detects which cipher to use, based on the ciphertext.



    


    
      
        decrypt!(ciphertext)

      


        Like decrypt/1, but raises any errors.



    


    
      
        encrypt(plaintext)

      


        Encrypts a binary using the first configured cipher in the vault's
configured :ciphers list.



    


    
      
        encrypt(plaintext, label)

      


        Encrypts a binary using the vault's configured cipher with the
corresponding label.



    


    
      
        encrypt!(plaintext)

      


        Like encrypt/1, but raises any errors.



    


    
      
        encrypt!(plaintext, label)

      


        Like encrypt/2, but raises any errors.



    


    
      
        json_library()

      


        The JSON library the vault uses to convert maps and lists into
JSON binaries before encryption.

      Link to this callback

          @callback decrypt(ciphertext()) :: {:ok, plaintext()} | {:error, Exception.t()}


      


Decrypts a binary with the configured cipher that generated the binary.
Automatically detects which cipher to use, based on the ciphertext.

  



  
    
      
      Link to this callback

          @callback decrypt!(ciphertext()) :: plaintext() | no_return()


      


Like decrypt/1, but raises any errors.

  



  
    
      
      Link to this callback

          @callback encrypt(plaintext()) :: {:ok, ciphertext()} | {:error, Exception.t()}


      


Encrypts a binary using the first configured cipher in the vault's
configured :ciphers list.

  



  
    
      
      Link to this callback

          @callback encrypt(plaintext(), label()) :: {:ok, ciphertext()} | {:error, Exception.t()}


      


Encrypts a binary using the vault's configured cipher with the
corresponding label.

          @callback encrypt!(plaintext()) :: ciphertext() | no_return()


      


Like encrypt/1, but raises any errors.

  



  
    
      
      Link to this callback
    
    encrypt!(plaintext, label)


      
       
       View Source
     


  


  

      

          @callback encrypt!(plaintext(), label()) :: ciphertext() | no_return()


      


Like encrypt/2, but raises any errors.

  



  
    
      
      Link to this callback
    
    json_library()


      
       
       View Source
     


  


  

      

          @callback json_library() :: module()


      


The JSON library the vault uses to convert maps and lists into
JSON binaries before encryption.

  


        

      



  

    
Cloak.Ciphers.AES.CTR 
    



      
A Cloak.Cipher which encrypts values with the AES cipher in CTR (stream) mode.
Internally relies on Erlang's :crypto.stream_encrypt/2.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions
  


    
      
        can_decrypt?(ciphertext, opts)

      


        Callback implementation for Cloak.Cipher. Determines if
a ciphertext can be decrypted with this cipher.



    


    
      
        decrypt(ciphertext, opts)

      


        Callback implementation for Cloak.Cipher. Decrypts a value
encrypted with AES in CTR mode.



    


    
      
        encrypt(plaintext, opts)

      


        Callback implementation for Cloak.Cipher. Encrypts a value using
AES in CTR mode.



    





      


      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    can_decrypt?(ciphertext, opts)


      
       
       View Source
     


  


  

Callback implementation for Cloak.Cipher. Determines if
a ciphertext can be decrypted with this cipher.

  



  
    
      
      Link to this function
    
    decrypt(ciphertext, opts)


      
       
       View Source
     


  


  

Callback implementation for Cloak.Cipher. Decrypts a value
encrypted with AES in CTR mode.
Uses the key tag to find the correct key for decryption, and the IV included
in the header to decrypt the body of the ciphertext.

  
  parameters

  
  Parameters


	ciphertext - Binary ciphertext generated by encrypt/2.


  
  examples

  
  Examples


iex> encrypt("Hello") |> decrypt
"Hello"

  



  
    
      
      Link to this function
    
    encrypt(plaintext, opts)


      
       
       View Source
     


  


  

Callback implementation for Cloak.Cipher. Encrypts a value using
AES in CTR mode.
Generates a random IV for every encryption, and prepends the key tag and IV to
the beginning of the ciphertext. The format can be diagrammed like this:
+-----------------------------------+----------------------+
|               HEADER              |         BODY         |
+-------------------+---------------+----------------------+
| Key Tag (n bytes) | IV (16 bytes) | Ciphertext (n bytes) |
+-------------------+---------------+----------------------+
|                   |__________________________________
|                                                     |
+---------------+-----------------+-------------------+
| Type (1 byte) | Length (1 byte) | Key Tag (n bytes) |
+---------------+-----------------+-------------------+
The Key Tag component of the header breaks down into a Type, Length,
and Value triplet for easy decoding.

  


        

      



  

    
Cloak.Ciphers.AES.GCM 
    



      
A Cloak.Cipher which encrypts values with the AES cipher in GCM (block) mode.
Internally relies on Erlang's :crypto.block_encrypt/4.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions
  


    
      
        can_decrypt?(ciphertext, opts)

      


        Callback implementation for Cloak.Cipher. Determines whether this module
can decrypt the given ciphertext.



    


    
      
        decrypt(ciphertext, opts)

      


        Callback implementation for Cloak.Cipher. Decrypts a value
encrypted with AES in GCM mode.



    


    
      
        encrypt(plaintext, opts)

      


        Callback implementation for Cloak.Cipher. Encrypts a value using
AES in GCM mode.



    





      


      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    can_decrypt?(ciphertext, opts)


      
       
       View Source
     


  


  

Callback implementation for Cloak.Cipher. Determines whether this module
can decrypt the given ciphertext.

  



  
    
      
      Link to this function
    
    decrypt(ciphertext, opts)


      
       
       View Source
     


  


  

Callback implementation for Cloak.Cipher. Decrypts a value
encrypted with AES in GCM mode.

  



  
    
      
      Link to this function
    
    encrypt(plaintext, opts)


      
       
       View Source
     


  


  

Callback implementation for Cloak.Cipher. Encrypts a value using
AES in GCM mode.
Generates a random IV for every encryption, and prepends the key tag, IV,
and ciphertag to the beginning of the ciphertext. The format can be
diagrammed like this:
+----------------------------------------------------------+----------------------+
|                          HEADER                          |         BODY         |
+-------------------+---------------+----------------------+----------------------+
| Key Tag (n bytes) | IV (n bytes)  | Ciphertag (16 bytes) | Ciphertext (n bytes) |
+-------------------+---------------+----------------------+----------------------+
|                   |_________________________________
|                                                     |
+---------------+-----------------+-------------------+
| Type (1 byte) | Length (1 byte) | Key Tag (n bytes) |
+---------------+-----------------+-------------------+
The Key Tag component of the header breaks down into a Type, Length,
and Value triplet for easy decoding.
Important: Because a random IV is used for every encryption, encrypt/2
will not produce the same ciphertext twice for the same value.

  


        

      



  

    
Cloak.Ciphers.Deprecated.AES.CTR 
    



      
DEPRECATED version of the Cloak.Ciphers.AES.CTR cipher, for use in
migrating existing data to the new format used by Cloak.Ciphers.AES.CTR.
Rationale
The old Cloak.AES.CTR cipher used the following format for ciphertext:
+---------------------------------------------------------+----------------------+
|                         HEADER                          |         BODY         |
+----------------------+------------------+---------------+----------------------+
| Module Tag (n bytes) | Key Tag (1 byte) | IV (16 bytes) | Ciphertext (n bytes) |
+----------------------+------------------+---------------+----------------------+
The new Cloak.Ciphers.AES.CTR implementation no longer prepends the "Module Tag"
component, and uses a new format as described in its docs. This cipher can
assist in upgrading old ciphertext to the new format.
See the Upgrading from 0.6.x guide for usage.

      





  

    
Cloak.Ciphers.Deprecated.AES.GCM 
    



      
DEPRECATED version of the Cloak.Ciphers.AES.GCM cipher, for use in
migrating existing data to the new format used by Cloak.Ciphers.AES.GCM.
Rationale
The old Cloak.AES.GCM cipher used the following format for ciphertext:
+---------------------------------------------------------+----------------------+
|                         HEADER                          |         BODY         |
+----------------------+------------------+---------------+----------------------+
| Module Tag (n bytes) | Key Tag (1 byte) | IV (16 bytes) | Ciphertext (n bytes) |
+----------------------+------------------+---------------+----------------------+
The new Cloak.Ciphers.AES.GCM implementation no longer prepends the "Module Tag"
component, and uses a new format as described in its docs. This cipher can
assist in upgrading old ciphertext to the new format.
See the Upgrading from 0.6.x guide for usage.

      





  

    
Cloak.InvalidConfig exception
    



      
Raised when there is invalid configuration.

      





  

    
Cloak.MissingCipher exception
    



      
Indicates that no cipher was found to encrypt/decrypt a given text.

